

# Introduction to Prolog

## 1 Objectives

In this laboratory session you will get familiar with the principal concepts of the Prolog language: facts, rules and queries. Additionally, the following will be presented: data types used in Prolog and unification rules.

### 1.1 Uses of Prolog

Prolog was created in 1972 by Alain Colmerauer and Philippe Roussel, based on the procedural interpretation of Horn clauses. Its popularity picked up in the 80's and 90's throughout Europe, USA and Japan for different systems. It was most used in the development of a particular set of application areas, natural language processing, expert systems (animal care - Auspig, water distribution - WADNES and SERPES, sport advisor - Perfect Pitch) and environmental systems (weather predictions - MM4 Weather Modeling System). The reason for this upsurge was the fast incremental development cycle and rapid prototyping capabilities in the solving of AI problems. Later, its appeal was increased by Object-oriented extensions. Other areas of use include: theorem proving, type systems, automated planning and its originally intended use, natural language processing.

Nowadays, some of the applications of Prolog can be found in (industrial, medical & commercial areas):

- expert systems that solve complex problems without the help of humans (e.g. automatically planning, monitoring, controlling and troubleshooting complex systems)
- decision support systems aiding organizations in decision-making (e.g. decision systems for medical diagnoses)
- online support service for customers

Notably, Prolog has been used in a question-answering computer system capable of answering questions posed in natural language, called Watson (developed by IBM). Specifically, Prolog was used for pattern matching on natural language parse trees.

## 1.2 Why should you learn Prolog?

Learning Prolog is highly beneficial for students due to several reasons. Firstly, it introduces them to the logic programming paradigm, offering a unique perspective on problem-solving distinct from conventional approaches. This fosters a deeper understanding of logic and deductive reasoning, essential skills applicable across various domains. Its support for constraint logic programming further enriches problem-solving abilities, particularly in constraint satisfaction problems. Moreover, Prolog's prominence in artificial intelligence (AI) and expert systems exposes students to cutting-edge technologies and prepares them for roles in AI research and development. Prolog also facilitates ML explainability by providing a transparent and rule-based approach to problem-solving, aiding in understanding and interpreting machine learning models. With Prolog, students embrace a new thought paradigm, where solutions are derived from logical rules and facts rather than explicit instructions. Moreover, Prolog's support for recursion makes it easy to understand and implement recursive algorithms, reinforcing concepts of recursion learned in data structures and algorithms courses. Its interpreter-based nature allows for instantly runnable and testable code, providing immediate feedback and facilitating iterative development. Finally, learning Prolog not only enhances students' problem-solving skills but also offers a recap of data structures and algorithms in a new and engaging paradigm.

## 2 Theoretic Considerations

### 2.1 The Logic Paradigm

A programming paradigm is a fundamental style of programming that dictates:

- The representation of data (ex: facts, variables, classes)
- The preprocessing of data (ex: assignments, comparisons, evaluations)

The *Logic Paradigm* belongs to the *Declarative Paradigm*. A declarative programming language answers the „**WHAT** should a program do” question. We will consider the following example in the SQL declarative language:

```
SELECT last_name, first_name FROM Students WHERE year = 3
```

The above instructions do not tell the SQL interpreter how to achieve the search, it only tells the interpreter what conditions the result must respect. Prolog is a logic programming language, therefore it is also a declarative one.

In contrast, an imperative programming language (ex: C, Pascal, C++, C#, Java etc.) answer the „**HOW** should a program solve a problem” question.

## 2.2 Prolog Concepts

Prolog instructions are represented by *facts* and *rules*. Once the Prolog source program has been consulted by the interpreter, it allows the asking of questions – through *queries*. The answering of questions is determined on the basis of the *facts* and *rules* through the use of the *backtracking* technique.

Anything that is unknown or cannot be proven is considered false.

Comments:

- one line comments start with the % symbol
- multiple line comments will be framed by the /\* and \*/ symbols (similar to comments in the C programming language)

## 2.3 Data Types

Prolog's single data type is the term. Terms are either: atoms, numbers, variables or compound terms (structures). The figure below presents a classification of the data types in Prolog:




Figure 1 – Prolog data types<sup>1</sup>

Simple:

- constants
  - numbers (ex: 47, 6.3 etc.)
    - In contrast to C – Prolog does not require a data type declaration as ,int' or ,float'
  - symbols/atoms (ex: girafa, 'Romania', 'antilopa Gnu' etc.)
    - If they contain special characters they need to be surrounded by simple quotation marks
- variables
  - Start with **Uppercase letter** or with **\_** (ex: X, Aux, \_12 etc.)
  - A single **\_** represents a free/anonymous variable

Compound:

- structures (ex: t(1, t(-2, nil, nil), t(8,nil,nil)) etc.)
  - lists (ex: [], [1, 2, 3], [1, 2 | \_] etc.)
    - [] = empty list
    - The list [1,2,3] is internally represented through '.(1, '.(2, '.(3, [])))
    - The template **[H|T]** separates the list in H (= the head of the list) and T (= the tail, **a list** that contains the rest of the elements)
  - strings (ex: "Hello World")

### 2.3.1 Exercises on Data Types

1. Which is the nature of the following Prolog terms:

- |         |            |                       |
|---------|------------|-----------------------|
| a. X    | d. hello   | g. [a, b, c]          |
| b. 'X'  | e. Hello   | h. [A, B, C]          |
| c. _138 | f. 'Hello' | i. [Ana,has,'apples'] |

2. Look up the following built-in predicates:

- var(Term)
- nonvar(Term)
- number(Term)
- atom(Term)
- atomic(Term)

*Note.* Prolog defines the *atomic/1* predicate as True if the term is instantiated (i.e. not a variable) and not compound. Thus, atoms and numbers are considered atomic. Moreover, the empty list [] is also considered atomic, as it is the base of the compound structure of the list.

### 2.3.2 Additional information about the List

Knowing that the template **[H|T]** can be used to separate a list into its head and its tail then (*Remember:* the tail is a list as well):

[1,2,3] can be written as [1 | [2,3] ].

*Question:* Using the template is the way to traverse a list, but what programming technique should be used?

*Answer:* Recursion. The head is separated from the tail and a recursion is applied on the tail, allowing the processing of the head.

Viewing the template in a recursive fashion, it can again be written as:

[1 | [2 | [3 | [] ] ] ]

## 2.4 Facts

Facts are predicates that are **always true**. They are also called axioms (from mathematics). They are allowed to have zero, one or more arguments. The **arity** of a predicate represents the number of arguments of said predicate.

Examples:

```
we_are_in_classroom_108.  
animal(elephant).  
animal('Gnu antilope').  
height(girafa, 5.5). % height in metres  
tree( t(1, t(-2, nil, nil), t(8,nil,nil)) ).
```

## 2.5 Queries

Queries can be viewed as the goals of the Prolog program. The answer of a query can be affirmative or negative. If we use variables in the question we can obtain other information as well.

Examples:

```
?- we_are_in_classroom_108.  
true.  
  
?- animal('Gnu antilope').  
true.  
?- animal(X).  
X = elephant; % repeat query with ; or n or space  
X = 'Gnu antilope';  
false. % there are no other animals defined in the program
```

## 2.6 Unification

The = symbol realizes the unification between the 2 terms.

*Value* is any data type that is not / does not contain an uninstantiated variable.

| Term 1                         | Term 2                         | Operation                                                                      |
|--------------------------------|--------------------------------|--------------------------------------------------------------------------------|
| Value (/instantiated variable) | Value (/instantiated variable) | Comparison                                                                     |
| Value (/instantiated variable) | Uninstantiated variable        | Assignment                                                                     |
| Uninstantiated variable        | Value (/instantiated variable) | Assignment                                                                     |
| Uninstantiated variable        | Uninstantiated variable        | Variables become synonymous<br>(as in pointing to the same location in memory) |

*Note.* Do not confuse the unification of Prolog with the assignment of C.

### 2.6.1 Exercises on Unification

1. Execute the following unification queries:

- a.  $?- a = a.$
- b.  $?- a = b.$
- c.  $?- 1 = 2.$
- d.  $?- 'ana' = 'Ana'.$
- e.  $?- X = 1, Y = X.$
- f.  $?- X = 3, Y = 2, X = Y.$
- g.  $?- X = 3, X = Y, Y = 2.$
- h.  $?- X = ana.$
- i.  $?- X = ana, Y = 'ana', X = Y.$
- j.  $?- a(b,c) = a(X,Y).$
- k.  $?- a(X,c(d,X)) = a(2,c(d,Y)).$
- l.  $?- a(X,Y) = a(b(c,Y),Z).$
- m.  $?- tree(left, root, Right) = tree(left, root, tree(a, b, tree(c, d, e))).$
- n.  $?- k(s(g),t(k)) = k(X,t(Y)).$
- o.  $?- father(X) = X.$
- p.  $?- loves(X,X) = loves(marsellus,mia).$
- q.  $?- [1, 2, 3] = [a, b, c].$
- r.  $?- [1, 2, 3] = [A, B, C].$
- s.  $?- [abc, 1, f(x) \mid L2] = [abc \mid T].$
- t.  $?- [abc, 1, f(x) \mid L2] = [abc, 1, f(x)].$

*Note1.* A list of format  $[1,2,3]$  is actually equivalent to  $[1,2,3 \mid []]$  and to  $[1 \mid [2 \mid [3 \mid []]]]$  as well.

*Note2.* The **[H|T]** template can be extended to  $[H1, H2 \mid T]$  or  $[H1, H2, H3 \mid T]$  or even more. It is important to *remember* that the head cannot be empty, while the tail can be an empty list  $[]$ . This brings about the implication that a template of type  $[H1, H2, \dots, Hn \mid T]$  can only unify with a list of minimum  $n$  elements. This means that there must be as many stopping conditions as there are heads in order to address all possible cases.

*Note\*.* (Remember) The unification of Prolog and the assignment of C work in different ways even though they use the same symbol  $=$ . While the assignment of C works in a unidirectional manner, the unification of prolog is bidirectional (as the order of the terms does not matter). In some cases it is easier to view it

as similar to a comparison (as it even returns a true/false), though this is not an entirely correct view either.

## 2.7 Rules

The rule represents the definition of a predicate under the form of a clause of Horn type:

$p(\dots)$  if  $p_{11}(\dots)$  and  $p_{12}(\dots)$  and ... and  $p_{1n}(\dots)$ .

...

$p(\dots)$  if  $p_{m1}(\dots)$  and  $p_{m2}(\dots)$  and ... and  $p_{mk}(\dots)$ .

$p(\dots)$  = the head of the clause (one predicate at maximum)

$p_{11}(\dots)$  and ... = the body of the clause (zero, one or more predicates)

fact = clause without a body

query = clause without a head

Special symbols (operators) in Prolog:

`:-` = "if"

`,` = "and"

`;` = "or" (can be achieved - and is equivalent - through the writing of multiple clauses of the same predicate)

Examples:

```
taller(X,Y) :- height(X,Hx), height(Y,Hy), Hx>Hy.  
% first we take the height of X, then we take the height of Y  
% and finally we compare the heights through the inequality  
  
path(X,Y) :- edge(X,Y).  
path(X,Y) :- edge(X,Intermediary), path(Intermediary,Y).  
% the path between nodes X and Y can be a direct connection between  
% the 2 nodes OR if no direct connection exists, we search for a  
% indirect connection through an intermediary node
```



## 2.8 SWISH vs SWI

During the laboratory sessions, there are two options of interpreters:

### 2.8.1 (Option1) SWI Prolog (requires installing) & Notepad++


The source program is written in Notepad++ as a \*.pl file and will be consulted into the SWI Prolog interpreter using its menu bar *File -> Consult*. Subsequent consultations of the same file can be surpassed through the use of *File -> Reload modified files*.

### 2.8.2 (Option2) SWISH Prolog – online platform (no install required)

In contrast, the SWISH Prolog platform incorporates both the editor and the interpreter and allows the use of both through a simple GUI.

### 2.8.3 Setup

This first image presents the SWI Prolog & Notepad ++ setup.



This second image presents the SWISH Prolog site. The colors of the frames have been adjusted to be the corresponding parts of the two options.

SWISH -- SWI-Prolog for Sharing x +

penguins.swi-prolog.org/apps/swish/

SWISH File Edit Examples Help

58 users online Search

Program

```
1 % Ex. We will be using the all-known statement of Socrates
2 % 'All men are mortal'
3
4 mortal(socrates). %(fact) (if not => error, human undefined procedure)
5
6 human(X) :- mortal(X). %(rule)
```

human(socrates).

true

?- human(socrates).


Examples History Solutions

table results Run!

## 2.9 Debugging in Prolog

Debugging in Prolog is called tracing and "*trace*" is the Prolog command that allows the debugging of Prolog code.

The SWI version of tracing is presented in the following figure:



The screenshot shows the SWI-Prolog environment. On the left, a text editor displays a Prolog program:

```
1 % Ex. We will be using the all-known statement of Socrates
2 % 'All men are mortal'
3
4 mortal(socrates). %(fact) (if not => error, human undefined procedure)
5
6 human(X) :- mortal(X). %(rule)
7
8
```

On the right, the SWI-Prolog console shows the following output:

```
Welcome to SWI-Prolog (threaded, 64 bits, version 8.4.2)
SWI-Prolog comes with ABSOLUTELY NO WARRANTY. This is free software.
Please run ?- license. for legal details.

For online help and background, visit https://www.swi-prolog.org
For built-in help, use ?- help(Topic). or ?- apropos(Word).

?-
% c:/Users/Rici/Desktop/ShowOff.pl compiled 0.00 sec, 2 clauses
?- human(socrates).
true.

?- trace.
true.

[trace] ?- human(socrates).
Call: (10) human(socrates) ? creep
Call: (11) mortal(socrates) ? creep
Exit: (11) mortal(socrates) ? creep
Exit: (10) human(socrates) ? creep
true.


[trace] ?- notrace.
true.

[debug] ?- human(socrates).
true.

[debug] ?-
```

It requires the enabling of trace through the running of the "*trace*" command beforehand. This can be stopped through the use of the "*notrace*" command.

The SWISH version of tracing:



The screenshot shows the SWISH web interface. On the left, a text editor displays a Prolog program:

```
1 % Ex. We will be using the all-known statement of Socrates
2 % 'All men are mortal'
3
4 mortal(socrates). %(fact) (if not => error, human undefined procedure)
5
6 human(X) :- mortal(X). %(rule)

```

On the right, the SWISH console shows the following output:

```
human(socrates).
true

trace, human(socrates).
Call: human(socrates)
Call: mortal(socrates)
Exit: mortal(socrates)
Exit: human(socrates)
true

?- trace, human(socrates).
```

The SWISH version requires the writing of "*trace*" before the predicate that will be run subsequently.

## 2.10 Backtracking

An intrinsic functionality of Prolog is backtracking. Query Answering in Prolog can be viewed (if simplified) as a Depth-First Search (DFS) process. When running a query the interpreter goes into an in-depth search for an answer, if the query is repeated (through ; or Next) the backtracking process starts looking for *another* answer. Remember, it needs to be a distinct answer, otherwise it would result in an infinite loop of the same answer.

Let's implement the following code that determines who can hold a party depending on a list of fanciful requirements:

```
% hold_party/1 is a rule that depends upon the successful execution
% of the birthday/1 and happy/1 facts
hold_party(X):-
    birthday(X),
    happy(X).

% a series of birthday/1 and happy/1 facts
birthday(alex).
birthday(maria).
birthday(adriana).
happy(ana).
happy(george).
happy(adriana).
```

*Note1.* Remember "/x" represents the arity of a predicate = the number of parameters.

*Note2.* In general, we want facts to be specific -> as you can see, they contain constants. Simultaneously, we want rules to be general -> they use a variable. Why? When we query a rule, we want to be able to let it find an answer however vague the question is. If the rule would be specific, we would only find one answer.

We will be following the execution of the *hold\_party/1* predicate through SWISH Prolog:

The screenshot shows the SWISH Prolog environment. On the left, a program is defined with the following clauses:

```

1 hold_party(X):-
2   birthday(X),
3   happy(X).
4
5 birthday(alex).
6 birthday(maria).
7 birthday(adriana).
8 happy(ana).
9 happy(george).
10 happy(adriana).
11

```

On the right, the execution trace for the query `trace, hold_party(X).` is displayed. The trace shows the following steps:

```

Call: hold_party(_4082)
Call: birthday(_4082)
Exit: birthday(alex) } - unifies with the first birthday/1 fact it finds -> X will unify with 'alex'
Call: happy(alex)
Fail: happy(alex) } - searches to see if happy(alex) exists, as it doesn't -> it fails
Redo: birthday(_4082) } - 'Redo' specifies the backtracking, it searches for a distinct birthday/1 fact and finds maria
Call: happy(maria)
Fail: happy(maria) } - Again it fails, as there is no happy(maria) fact
Redo: birthday(_4082) } - The final backtracking, it unifies with birthday(adriana) -> X=adriana
Call: happy(adriana)
Exit: happy(adriana) } - Verifies if happy(adriana) exists - and it does, exits with success
Exit: hold_party(adriana) -> exits the hold_party call and returns answer
X = adriana

```

At the bottom, the query `?- trace, hold_party(X).` is entered, and the `Run!` button is visible.

## 2.11 Recursion

Let's empirically separate recursion into a subset of fundamental parts, we will be using the C implementation of factorial:

1. Any recursion requires, by its definition, the call of the function within the function.
2. It also contains a step, in the case of the factorial, the step is a decrement (or increment).
3. Then, there's the processing – for the factorial, this is the multiplication – this can be overlooked for now.
4. Let's follow what happens when we try to compute the factorial of 3.
  - a. Multiply by 3 & decrement -> recursion
  - b. Multiply by 2 & decrement -> recursion
  - c. Multiply by 1 & decrement -> recursion
  - d. Multiply by 0 & decrement -> recursion
  - e. Multiply by -1 & decrement -> recursion
  - f. ...
  - g. Overflow – Multiply by 2147483647 & decrement -> recursion
  - h. *Do you see the problem?* Ignoring the fact that there's a multiplication by 0, it's an infinite loop. An integral part of a recursion call is the stopping condition.

We will be taking the example of a student that has been at home after the exam session and is trying to get back to school.

```

% the on_route/1 fact
on_route(dorms).
% the on_route/1 rule - this is a recursive rule
on_route(Place):-
    move(Place, Method, NewPlace),
    on_route(NewPlace).

% the move/3 facts
move(home, taxi, trainstation).
move(trainstation, train, cluj).
move(cluj, bus, dorms).

```

We will be following the execution of the *on\_route/1* predicate through SWI-Prolog:

```

C:\Users\ardel\Desktop\on_route.pl - Notepad++
File Edit Search View Encoding Language Settings Tools Macro Run Plugins Window
?
on_route.pl
1 % the on_route/1 fact
2 on_route(camin).
3 % the on_route/1 rule - this is a recursive rule
4 on_route(Place):-
5     move(Place, Method, NewPlace),
6     on_route(NewPlace).
7
8 % the move/3 facts
9 move(acasa, taxi, gara).
10 move(gara, tren, cluj).
11 move(cluj, bus, camin).
12

SWI-Prolog (AMD64, Multi-threaded, version 8.4.2)
File Edit Settings Run Debug Help
Warning: c:/users/ardel/desktop/on_route.pl:4:
Warning: Singleton variables: [Method]
% c:/Users/ardel/Desktop/on_route.pl compiled 0.00 sec, 5 clauses
?- trace.
true.

[trace] ?- on_route(acasa).
Call: (10) on_route(acasa) ? creep
Call: (11) move(acasa, _57006, _56946) ? creep % Place unified with acasa (line 4) -> calls line 5
Exit: (11) move(acasa, taxi, gara) ? creep % it searches for a move with the
Call: (11) on_route(gara) ? creep % first paramter = acasa and finds the first fact
Call: (12) move(gara, _59274, _59214) ? creep % NewPlace unified with gara and recursion (line 6) is called
Exit: (12) move(gara, tren, cluj) ? creep % Through recursion we are back at line 4
Call: (12) on_route(cluj) ? creep % and calls line 5 with Place=gara
Call: (13) move(cluj, _61542, _61482) ? creep
Exit: (13) move(cluj, bus, camin) ? creep
Call: (13) on_route(camin) ? creep
Exit: (13) on_route(camin) ? creep
Exit: (12) on_route(cluj) ? creep
Exit: (11) on_route(gara) ? creep
Exit: (10) on_route(acasa) ? creep
true.

[trace] ?- |

```

Note1. Do not forget to File->Consult it will result in the green comment.

Note2. Do not forget to activate "trace".

### 3 Final Exercise

1. Write new predicates for the kinship relations.




Figure 1. Family tree

```
% The woman/1 predicate - remember /x specifies the arity as ,x'
woman(ana).
woman(sara).
woman(ema).
woman(maria).
% ... add the remaining facts of this predicate

% The man/1 predicate
man(andrei).
man(george).
man(alex).
% ... add the remaining facts of this predicate

% The parent/2 predicate
parent(maria, ana). % maria is the parent of ana
parent(george, ana). % george is the parent of ana
parent(maria, andrei).
parent(george, andrei).
% ... add the remaining facts of this predicate

% The mother/2 predicate - based on the parent and woman predicates
% X is the mother of Y, if X is a woman and X is the parent of Y
mother(X,Y) :- woman(X), parent(X,Y).
```

*For SWI-Prolog (the installable version):*

The source program (ex: genealogy.pl) can be consulted through the menu bar of the interpreter: *File -> Consult* or it can be written in the interpreter through the use of the following built-in predicate:

```
?- consult('C:/Users/student/Desktop/genealogy.pl').  
true. % If no error intervenes, it will return true
```

1.1. Test the following queries:

*\*Do remember: queries are preceded by the '?' operator, which is already written, the rest of the lines represent the answers of Prolog to the queries):*

```
?- man(george).    % is george a man?  
true.
```

```
?- man(X).         % who is a man?
```

```
X = andrei ? ;    % we can repeat the question using ; or n or space
```

```
X = george ? ;
```

```
X = alex.
```

```
?- parent(X, andrei). % Who are the parents of andrei?
```

```
X = maria ? ;
```

```
X = george.
```

```
?- parent(maria, X). % Who are the children of maria?
```

```
X = ana ? ;
```

```
X = andrei.
```

```
?- mother(ana, X). % Who are the children of ana?
```

```
false.
```

```
?- mother(X, ana). % Who is the mother of ana?
```

```
X = maria ? ; % repeat question: Does ana have another mother?
```

```
false.
```



- 1.2. Write the father/2 predicate.
- 1.3. Complete the man/1, woman/1 and parent/2 predicates to cover the whole genealogic tree from above.

1.4. Test the following queries:

?- father(alex, X).  
?- father(X, Y).  
?- mother(dorina, maria).

1.5. Test the following predicates:

**% The sibling/2 predicate**  
**% X and Y are siblings if they have at least one parent in common**  
**% and X is different from Y**  
sibling(X,Y) :- parent(Z,X), parent(Z,Y), X\=Y.

**% The sister/2 predicate**  
**% X is the sister of Y if X is a woman and X and Y are siblings**  
sister(X,Y) :- sibling(X,Y), woman(X).

**% The aunt/2 predicate**  
**% X is the aunt of Y if she(X) is the sister of Z and Z is the parent of Y**  
aunt(X,Y) :- sister(X,Z), parent(Z,Y).

- 1.6. Write the brother/2, uncle/2, grandmother/2 and grandfather/2 predicates.
- 1.7. Follow the steps of finding the answer of the following queries (through the use of *trace*):

?- aunt(carmen, X).  
?- grandmother(dorina, Y).  
?- grandfather(X, ana).

- 1.8. Write the ancestor/2 predicate. X is the ancestor of Y if X is linked to Y through a series (regardless of number) of parent relationships.